
Sarcina lucrării:
Sarcina acestei lucrări de laborator este crearea unei aplicații ce va rula în Excel și va folosi un șir de funcții definite de programator destinate prelucrării vectorilor în Excel. 

Va fi creată o aplicație care calculează suma elementelor unui vector, suma a 2 vectori, va afla elementul minim și maxim al vectorului, va scalariza un vector și va sorta un vector. Toate aceste acțiuni vor fi îndeplinite cu ajutorul funcțiilor definite de către programator.
Codul sursă al modulului cu funcții definite:

	Public Function SumaVector(rng As Range) As Double

    Dim i As Integer

    Dim n As Integer

    Dim Suma As Double

    n = rng.Rows.Count

    ReDim vect(1 To n) As Double

    Suma = 0

    For i = 1 To n

        vect(i) = rng(i)

        Suma = Suma + vect(i)

    Next i

    SumaVector = Suma

End Function

Public Function MaxVector(rng As Range) As Double

    Dim i As Integer

    Dim n As Integer

    Dim max As Double

    ' Preluarea plajei de celule

    n = rng.Rows.Count

    ReDim vect(1 To n) As Double

    For i = 1 To n: vect(i) = rng(i): Next i

    max = vect(1)

    For i = 2 To n

        If (vect(i) > max) Then

            max = vect(i)

        End If

    Next i

    MaxVector = max

End Function

Public Function MinVector(rng As Range) As Double

    Dim i As Integer

    Dim n As Integer

    Dim min As Double

    ' Preluarea plajei de celule

    n = rng.Rows.Count

    ReDim vect(1 To n) As Double

    For i = 1 To n: vect(i) = rng(i): Next i

    min = vect(1)

    For i = 2 To n

        If (vect(i) < min) Then

            min = vect(i)

        End If

    Next i

    MinVector = min

End Function

Public Function ScaleVector(rng As Range, sc As Double) As Double()

    Dim i As Integer

    Dim n As Integer

    ' Preluarea plajei de celule

    n = rng.Rows.Count

    ReDim vect(1 To n) As Double

    For i = 1 To n: vect(i) = rng(i): Next i

    ReDim scaled(1 To n) As Double

    For i = 1 To n

        scaled(i) = sc * vect(i)

    Next i

    ScaleVector = scaled

End Function

Public Function Suma2Vec(rng1 As Range, rng2 As Range) As _

Variant

    Dim i As Integer

    Dim n1 As Integer

    Dim n2 As Integer

    ' Preluarea plajei de celule

    n1 = rng1.Rows.Count

    n2 = rng2.Rows.Count

    If (n1 <> n2) Then

        Suma2Vec = "Dimensiuni inegale."

        Exit Function

    End If

    ReDim vect(1 To n1) As Double

    For i = 1 To n1

        vect(i) = rng1(i) + rng2(i)

    Next i

    Suma2Vec = vect

End Function

Public Function Bubble(rng As Range) As Double()

    Dim temp As Double

    Dim i As Integer, j As Integer

    Dim n As Integer

    n = rng.Rows.Count

    ReDim A(1 To n) As Double

    For i = 1 To n

        A(i) = rng(i)

    Next i

    For i = 1 To n - 1

        For j = 1 To n - i

            If (A(j) > A(j + 1)) Then

                temp = A(j)

                A(j) = A(j + 1)

                A(j + 1) = temp

            End If

        Next j

    Next i

    Bubble = A

End Function


Codul sursă pentru forma creată:

	Dim vect As Range

Private Sub cmdExit_Click()

    MainUserForm.Hide

    Unload MainUserForm

End Sub

Private Sub cmdMinMax_Click()

    Dim min As Double

    Dim max As Double

    min = MinVector(vect)

    max = MaxVector(vect)

    viewMin.Caption = "Minimul este: " & min

    viewMax.Caption = "Maximul este: " & max

End Sub

Private Sub cmdScale_Click()

    Dim c As Double

    Dim n As Integer

    Dim col As Integer

    n = vect.Rows.Count

    c = insConst.Value

    ReDim scal(1 To n) As Double

    scal = ScaleVector(vect, c)

    col = insCol.Value

    Cells(1, col).Value = "Scaled"

    For i = 1 To n

        Cells(i + 1, col).Value = scal(i)

    Next i

End Sub

Private Sub cmdSort_Click()

    Dim col As Integer

    ReDim rez(1 To vect.Rows.Count) As Double

    rez = Bubble(vect)

    col = insCol4.Value

    Cells(1, col).Value = "Sorted"

    For i = 1 To vect.Rows.Count

        Cells(i + 1, col).Value = rez(i)

    Next i

End Sub

Private Sub cmdSuma_Click()

    Dim sum As Double

    sum = SumaVector(vect)

    viewSuma.Caption = "Suma este: " & sum

End Sub

Private Sub cmdSumeaza_Click()

    Dim vect1 As Range

    Dim vect2 As Range

    Dim rez As Variant

    Dim col As Integer

    Set vect1 = Range(insCol1.Value)

    Set vect2 = Range(insCol2.Value)

    rez = Suma2Vec(vect1, vect2)

    If (VarType(rez) = vbString) Then

        MsgBox rez

    Else

        col = insCol3.Value

        Cells(1, col).Value = "Suma 2 vectori"

        For i = 1 To vect1.Rows.Count

            Cells(i + 1, col).Value = rez(i)

        Next i

    End If

End Sub

Private Sub UserForm_Activate()

    Set vect = Worksheets("sheet1").Range(ActiveWindow.RangeSelection.Address)

End Sub


Descrierea aplicaţiei create:
Această formă reprezintă interfața de acces la aplicația creată.

Forma are mai mult pagini (tabs) de unde pot fi accesate diferite funcții. Pe pagina ”Suma vector” se află un buton la apăsarea căruia pentru vectorul selectat în Excel va fi calculată suma elementelor acestuia și afișată pe această pagină.

Pagina ”MinMax vector” conține un buton care pentru vectorul selectat află elemental minim și maxim și-l afișează pe această pagină.

Pagina Scale vector conține 2 cîmpuri de text unde utilizatorul trebuie să introducă constanta la care va fi înmulțit vectorul și numărul coloanei unde va fi inserat vectorul scalarizat. Apăsînd pe butonul ”Scale” în coloana indicată va fi inserat vectorul scalarizat.

Pe pagina ”Suma 2 vectori” sunt trei cîmpuri de text unde utilizatorul trebuie să introducă aria primului vectorul, aria vectorului II și numărul coloanei unde va fi inserat vectorul rezultat. În caz dacă vectorii au dimensiuni diferite va fi afișat un mesaj de avertizare.

Pe pagina ”Sortare vector” este un cîmp de text unde utilizatorul trebuie să introducă numărul coloanei unde va fi inserat vectorul sortat. Apăsînd pe butonul ”Sortare” se va produce sortarea.

Dacă lansăm aplicația creată obținem următoarele rezultate:

Sarcina individuală:

1. Elaboraţi forma propusă în lucrare.

2. Comentaţi detaliat codurile associate obiectelor formei propuse
3. Modificaţi forma şi codul respectiv pentru adăugarea cel puţin unui obiect/eveniment.
4. Faceţi concluzii
5. Prezentaţi lucrarea profesorului. 
După efectuarea lucrarii de laborator studentul trebuie să cunoască :
1. Cum se declara şi se apelează funcţiile utilizatorului  în VBA

2. Structura subrutinei şi instrucţiunile simple a limbajului VBA utilzate.
3. Tipurile de date simple  a limbajului VBA utilzate.
4.   Instucţiunile utilizate în codurile propuse
5.   Cum se adaugă o funcţie la un proiect

6. Cum se adugă un obiect nou pe obiectul Forma.
7. Cum se pot  modifica proprietaţile obiectelor la etapa de proiectare şi de executare. 

Autor: 


Examinator: 

